
*Set declaration
SETS
     skill             skilltypes of labor  /lowskilled, highskilled/
     ownership         private and public consumption   /private, public/
;

*Parameter definition
PARAMETERS
     phi               skilled and unskilled subst elasticity          /0.5/
     tm                marginal tax rate                              /0.26/
     ts                social security tax                     /0.321481274/
     theta             subst. elast. btw leisure and consumption         /2/
     kappa(ownership)  subst elast in cons btw imp-dom /private 2, public 2/
     xi                subst elast in foreign demand for imp-dom         /2/
     r                 discount rate                                   /0.1/
     rho               probability of getting caught                   /0.2/
     v                 exogeneous probability of quitting             /0.05/
     m(skill)          exogeneous parameter of labor supply
     productivity(skill) productivity parameter
     epsilon(skill)    inverse elasticity
     b                 parameter skilled and unskilled production
     q(ownership)      parameter in import and domestic consumption
     z                 parameter of import export
     alpha             employers power
     x                 parameter for efficiency wage
     d(skill)          parameter d to labor supply equation
     replacementrate(skill) replacement rate
     a(skill)          parameter for tax alloance
*Inital imported data
     aggr_production0
     employment0(skill)
     wage0(skill)
     aggr_profit0
     laborsupply0(skill)
     aggr_consumption0(ownership)
     exports0
     imp_consumption0(ownership)
     benefits0(skill)
     taxallowance0(skill)
*Prices which are normalised to one
     goodsprices0(skill)
     importsprice0
     consumptionprice0(ownership)
     productionprice0
*Initial variables which are calibrated from imported variables
     unemployment0(skill)
     dom_consumption0(ownership)

     production0(skill)
     profit0(skill)
     averagetax0(skill)
     goods_consumption0(skill)
     foreign_consumption0
;

*Data
    aggr_production0=60868;
    employment0("lowskilled")=142.1;
    employment0("highskilled")=435.6;
    wage0("lowskilled")=47.4102744546094;
    wage0("highskilled")=66.3636363636364;
    aggr_profit0=13764;
    laborsupply0("lowskilled")=175;
    laborsupply0("highskilled")=485.8;
    aggr_consumption0("public")=17661.156;
    aggr_consumption0("private")=43206.844;
    exports0=17593;
    imp_consumption0("private")=15588;
    imp_consumption0("public")=2005;
    benefits0(skill)=15.2;
    taxallowance0(skill)=12;

*Prices are initialized as ones
    goodsprices0("lowskilled")=1;
    goodsprices0("highskilled")=1;
    importsprice0=1;
    consumptionprice0(ownership)=1;
    productionprice0=1;

*Exogenously set parameter to some large value
    foreign_consumption0=1000000;

*Calibration
*1) Domestic consumption
     dom_consumption0(ownership)
     =aggr_consumption0(ownership)-imp_consumption0(ownership);

*2) Calibration of epsilon from 1-3
     epsilon(skill)=aggr_profit0/aggr_production0;

*3) Productivity
     productivity(skill)=wage0(skill)*(1+ts)/(1-epsilon(skill));

*4) Production
     production0(skill)=productivity(skill)*employment0(skill);

*5) aggr_profit
     profit0(skill)=goodsprices0(skill)*production0(skill)
     -wage0(skill)*(1+ts)*employment0(skill);

*6) Calibration of b (skilled and unskilled production)
*!!calibrated for phi=0.5, therefore 2 in the equation
     b=(2*aggr_production0*production0("highskilled")
     -2*((aggr_production0*(production0("lowskilled")**2)
     *production0("highskilled"))
     -((aggr_production0**2)*production0("lowskilled")
     *production0("highskilled"))
     +(aggr_production0*(production0("highskilled")**2)
     *production0("lowskilled")))**(1/2))
     /(2*(aggr_production0*production0("lowskilled")
     +aggr_production0*production0("highskilled")));

*7) Calibration of q in import and domestic consumption
     q(ownership)=(aggr_consumption0(ownership)*imp_consumption0(ownership)
     +imp_consumption0(ownership)*dom_consumption0(ownership)

     +(dom_consumption0(ownership)**2)
     -dom_consumption0(ownership)*aggr_consumption0(ownership)
     +2*(aggr_consumption0(ownership)*(imp_consumption0(ownership)**2)
     *dom_consumption0(ownership)
     -(aggr_consumption0(ownership)**2)*imp_consumption0(ownership)
     *dom_consumption0(ownership)
     +aggr_consumption0(ownership)*imp_consumption0(ownership)
     *dom_consumption0(ownership)**2)**(1/2))
     /(2*imp_consumption0(ownership)*dom_consumption0(ownership)
     +(imp_consumption0(ownership)**2)+dom_consumption0(ownership)**2);

*8) Calibration of m
     m(skill)=laborsupply0(skill)*3/2;

*9) Parameter in tax allowance
     a(skill)=taxallowance0(skill)/wage0(skill);

*10) Average tax rate
     averagetax0(skill)=tm-tm*taxallowance0(skill)
     *laborsupply0(skill)/(wage0(skill)*laborsupply0(skill));

*11) Employers bargaining power alpha
     alpha=(wage0("lowskilled")*epsilon("lowskilled")
     /(2*(1-epsilon("lowskilled"))))/((wage0("lowskilled")
     *((1-averagetax0("lowskilled"))/(1-tm)))
     +(wage0("lowskilled")*epsilon("lowskilled")
     /(2*(1-epsilon("lowskilled"))))
     -(benefits0("lowskilled")/(1-tm)));

*12) Replacement rates
     replacementrate(skill)=benefits0(skill)
     /(0.5*(wage0("lowskilled")+wage0("highskilled")));

*13) Unemployment rate
     unemployment0(skill)=(laborsupply0(skill)
     -employment0(skill))/laborsupply0(skill);

*14) Parameter in efficiency wage eq
     x=(wage0("highskilled")-benefits0("highskilled"))
     *unemployment0("highskilled")*rho/(unemployment0("highskilled")*rho
     +r*unemployment0("highskilled")+v);

*15) Parameter d in labor supply
     d(skill)=(-laborsupply0(skill)*wage0(skill)*(((1-tm)
     *wage0(skill)/consumptionprice0("private"))**(-theta)))*(-1
     +averagetax0(skill))/(-laborsupply0(skill)*consumptionprice0("private")
     +(laborsupply0(skill)*wage0(skill)*(((1-tm)
     *wage0(skill)/consumptionprice0("private"))**(-theta)))
     -((laborsupply0(skill)*wage0(skill)*(((1-tm)
     *wage0(skill)/consumptionprice0("private"))**(-theta)))
     *averagetax0(skill))+m(skill)*consumptionprice0("private"));

*16) Parameter z in export equation
     z=exports0/(foreign_consumption0
     *((productionprice0/importsprice0)**(-xi))+exports0);

*Creating tex file with the initial data report
*First creating a table with data
set  item1 /"Production", "Employment of unskilled",
     "Employment of skilled", "Wage of unskilled", "Wage of skilled",
     "aggr_profit", "Labor supply of unskilled", "Labor supply of skilled",
     "Unemployment of unskilled", "Unemployment of skilled", "Government",
     "Private consumption", "Export", "Imported private consumption",
     "Imported government consumption", "Benefit", "Tax allowance"/;

set      column1  /Data/;
parameter report1(item1, column1) "Data";

Page 3



HinnosaarMarit2004_EPfinal_web_gms.txt
   report1("Production", "Data")=aggr_production0;
   report1("Employment of unskilled", "Data")=employment0("lowskilled");
   report1("Employment of skilled", "Data")=employment0("highskilled");
   report1("Wage of unskilled", "Data")=wage0("lowskilled");
   report1("Wage of skilled", "Data")=wage0("highskilled");
   report1("aggr_profit", "Data")=aggr_profit0;
   report1("Labor supply of unskilled", "Data")=laborsupply0("lowskilled");
   report1("Labor supply of skilled", "Data")=laborsupply0("highskilled");
   report1("Unemployment of unskilled", "Data")=unemployment0("lowskilled");
   report1("Unemployment of skilled", "Data")=unemployment0("highskilled");
   report1("Government", "Data")=aggr_consumption0("public");
   report1("Private consumption", "Data")=aggr_consumption0("private");
   report1("Export", "Data")=exports0;
   report1("Imported private consumption","Data")=imp_consumption0("private");
   report1("Imported government 
consumption","Data")=imp_consumption0("public");
   report1("Benefit", "Data")=benefits0("lowskilled");
   report1("Tax allowance", "Data")=taxallowance0("lowskilled");

display  report1
* Generating tex file with the table:
file ktex /datafromgams.tex/; put ktex;
$setglobal format tex
$libinclude gams2tbl report1
putclose;

*Creating a table with calibrated parameters
set  item2 /"Subst. el.: skilled-unskilled",
     "Subst. el.: leisure-consumption", "private. subst. el.: imp.-dom",
     "public. subst. el.: imp.-dom", "Subst. el. in foreign demand",
     "Discount rate", "Prob. of getting caught", "Prob. of quitting a job",
     "Marginal tax rate", "Social security tax rate",
     "Productivity of unskilled",  "Productivity of skilled",
     "Inverse demand elast. uskilled", "Inverse demand elast. skilled",
     "Parameter: skilled-unskilled", "private. parameter: imp-dom",
     "public. parameter: imp-dom", "Employers bargaining power",
     "Efficiency wage", "Param. unskilled: leisure-cons.",
     "Param. skilled: leisure-cons.", "Replacement rate unskilled",
     "Replacement rate skilled"/;

parameter decimals(column1)  to report 4 decimals /"Data" 4/;
parameter report2(item2, column1) Parameters of the model;
     report2("Subst. el.: skilled-unskilled", "Data")=phi;
     report2("Subst. el.: leisure-consumption", "Data")=theta;
     report2("private. subst. el.: imp.-dom", "Data")=kappa("private");
     report2("public. subst. el.: imp.-dom", "Data")=kappa("public");
     report2("Subst. el. in foreign demand", "Data")=xi;
     report2("Discount rate", "Data")=r;
     report2("Prob. of getting caught", "Data")=rho;
     report2("Prob. of quitting a job", "Data")=v;
     report2("Marginal tax rate", "Data")=tm;
     report2("Social security tax rate", "Data")=ts;
     report2("Productivity of unskilled", "Data")=productivity("lowskilled");
     report2("Productivity of skilled", "Data")=productivity("highskilled");
     report2("Inverse demand elast. uskilled", "Data")=epsilon("lowskilled");
     report2("Inverse demand elast. skilled", "Data")=epsilon("highskilled");
     report2("Parameter: skilled-unskilled", "Data")=b;
     report2("private. parameter: imp-dom", "Data")=q("private");
     report2("public. parameter: imp-dom", "Data")=q("public");
     report2("Employers bargaining power", "Data")=alpha;
     report2("Efficiency wage", "Data")=x;
     report2("Param. unskilled: leisure-cons.", "Data")=d("lowskilled");
     report2("Param. skilled: leisure-cons.", "Data")=d("highskilled");
     report2("Replacement rate 
unskilled","Data")=replacementrate("lowskilled");
     report2("Replacement rate skilled","Data")=replacementrate("highskilled");

display  report2

* Then generating a tex file with the table:
$setglobal c_decimals decimals
file ltex /parametersfromgams.tex/; put ltex;
$setglobal format tex
$libinclude gams2tbl report2
putclose;

*Defining variables
POSITIVE VARIABLES
     production(skill)              domestic production
     aggr_production                domestic production
     profit(skill)                  aggr_profit
     aggr_profit                    aggregate profit
     wage(skill)                    wage
     employment(skill)              labor demand
     goodsprices(skill)             price
     laborsupply(skill)             labor supply
     aggr_consumption(ownership)    total private and public consumpion
     consumption_price(ownership)   price of consumption
     unemployment(skill)            unemployment rate of skill type
     importsprice                      price of imports
     productionprice                      price of domestic production
     imp_consumption(ownership)     private and public consumption of imports
     dom_consumption(ownership)     private and public consum of domestic goods
     exports                        exports
     averagetax(skill)              average aggregate tax rate
     benefits(skill)                benefit
     foreign_consumption            domestic production in foreign country
     taxallowance(skill)            tax alloance
;

*Defining a variable for GAMS program to solve
VARIABLES
     trick                     the dummy variable to be maximised for GAMS use
;

*Giving initial values to the variables
     production.l(skill)=production0(skill)                         ;
     aggr_production.l=aggr_production0                             ;
     profit.l(skill)=profit0(skill)                                 ;
     aggr_profit.l=aggr_profit0                                     ;
     wage.l(skill)=wage0(skill)                                     ;
     employment.l(skill)=employment0(skill)                         ;
     goodsprices.l(skill)=goodsprices0(skill)                       ;
     laborsupply.l(skill)=laborsupply0(skill)                       ;
     aggr_consumption.l(ownership)=aggr_consumption0(ownership)     ;
     consumption_price.l(ownership)=consumptionprice0(ownership)    ;
     unemployment.l(skill)=unemployment0(skill)                     ;
     importsprice.l=importsprice0                                   ;
     productionprice.l=productionprice0                             ;
     imp_consumption.l(ownership)=imp_consumption0(ownership)       ;
     dom_consumption.l(ownership)=dom_consumption0(ownership)       ;
     exports.l=exports0                                             ;
     averagetax.l(skill)=averagetax0(skill)                         ;
     benefits.l(skill)=benefits0(skill)                             ;
     foreign_consumption.l=foreign_consumption0                     ;
     taxallowance.l(skill)=taxallowance0(skill)                     ;

*Writing down equations
EQUATIONS
*Households:
     e_laborsupply(skill)         labor supply
     e_private_consumption        private consumption
     e_FOC_imports(ownership)     consumption mix of imported and domestic 
goods
     e_CES_imports                CES function of imported and domestic goods
     e_FOC_goods                  optimal commodity mix at domestic economy
     e_CES_goods                  aggregate production

*Firms
     e_production(skill)          production function
     e_price(skill)               optimal price
     e_profit(skill)              firm'highskilled aggr_profit
     e_aggr_profit                aggregate aggr_profit

*Labor market
     e_unionswage                 unions wage bargaining
     e_efficiencywage             efficiency wage
     e_unemployment(skill)        unemployment rate

*Prices
     e_domesticprices             aggregate price of domestic goods
     e_consumptionprices(ownership) aggregate price of consumption

*Government
     e_government                 government budget balance
     e_balance                    balance of payments

*Institutions
     e_benefits(skill)            benefits
     e_taxes(skill)               average tax
     e_taxallowance(skill)        tax alloance

*Exports demand
     e_trade                      trade

*Equilibrium
     e_walras                     equilibrium at the home market

*Objective
     objective                    objective function for GAMS
;

* Defining the equations
* Households
e_laborsupply(skill)..
     laborsupply(skill)=e=
         m(skill)/(1+((1-d(skill))/d(skill))*((1-averagetax(skill))
         *wage(skill)/consumption_price("private"))*(((1
         -tm)*wage(skill))/consumption_price("private"))**(-theta));

e_private_consumption..
     aggr_consumption("private")=e=
         (sum(skill,(((1-(tm-tm*taxallowance(skill)
         *employment0(skill)/(wage(skill)*employment(skill))))
         *wage(skill)*employment(skill))+(benefits(skill)*(laborsupply(skill)
         -employment(skill)))))+aggr_profit)/consumption_price("private");

e_FOC_imports(ownership)..
     imp_consumption(ownership)/dom_consumption(ownership)=e=
         q(ownership)/(1-q(ownership))
         *(importsprice/productionprice)**(-kappa(ownership));

e_CES_imports..
     aggr_consumption("private")=e=
         ((q("private")**(1/kappa("private")))
         *imp_consumption("private")**((kappa("private")-1)
         /kappa("private"))+((1-q("private"))**(1/kappa("private")))
         *dom_consumption("private")**((kappa("private")
         -1)/kappa("private")))**(kappa("private")/(kappa("private")-1));

e_FOC_goods..
     (production("highskilled")/production("lowskilled"))=e=
         (b/(1-b))*((goodsprices("highskilled")
         /goodsprices("lowskilled"))**(-phi));

e_CES_goods..
     aggr_production=e=
         ((b**(1/phi))*(production("highskilled")**((phi-1)/phi))
         +((1-b)**(1/phi))
         *(production("lowskilled")**((phi-1)/phi)))**(phi/(phi-1));

*Firms
e_production(skill)..
     production(skill)=e=
         productivity(skill)*employment(skill);

e_price(skill)..
     goodsprices(skill)=e=
         wage(skill)*(1+ts)/(productivity(skill)*(1-epsilon(skill)));

e_profit(skill)..
     profit(skill)=e=
         goodsprices(skill)*production(skill)
         -wage(skill)*(1+ts)*employment(skill);

e_aggr_profit..
     aggr_profit=e=
         sum(skill,profit(skill));

*Labor market
e_unionswage..
     wage("lowskilled")=e=
         (alpha*benefits("lowskilled")/(1-tm))/
         ((alpha*(1-averagetax("lowskilled"))/(1-tm))-(epsilon("lowskilled")/
         (1-epsilon("lowskilled")))*(1-alpha)/2);

e_efficiencywage..
     wage("highskilled")=e=
         benefits("highskilled")+x+(r+v/unemployment("highskilled"))*x/rho;

e_unemployment(skill)..
     unemployment(skill)=e=
         (laborsupply(skill)-employment(skill))/laborsupply(skill);

*Prices
e_domesticprices..
     productionprice=e=
         (b*goodsprices("highskilled")**(1-phi)
         +(1-b)*goodsprices("lowskilled")**(1-phi))**(1/(1-phi));

e_consumptionprices(ownership)..
     consumption_price(ownership)=e=
         (q(ownership)*importsprice**(1-kappa(ownership))+(1-q(ownership))
         *productionprice**(1-kappa(ownership)))**(1/(1-kappa(ownership)));

*Government
e_government..
     consumption_price("public")*aggr_consumption("public")=e=
         sum(skill,((tm-tm*taxallowance(skill)*employment0(skill)
         /(wage(skill)*employment(skill)))*wage(skill)*employment(skill)
         +ts*wage(skill)*employment(skill)-benefits(skill)*unemployment(skill)
         *laborsupply(skill)));

e_balance..
     productionprice*exports=e=
         importsprice*sum(ownership,imp_consumption(ownership));

*Institutions
e_benefits(skill)..
     benefits(skill)=e=
         replacementrate(skill)*0.5*(wage("lowskilled")+wage("highskilled"));

e_taxes(skill)..
     averagetax(skill)=e=
         tm-tm*taxallowance(skill)*laborsupply0(skill)
         /(wage(skill)*laborsupply(skill));

e_taxallowance(skill)..
     taxallowance(skill)=e=
         a(skill)*wage(skill);

*Export demand
e_trade..
     exports=e=
         (z/(1-z))*foreign_consumption*((productionprice/importsprice)**(-xi));

*Equilibrium
e_walras..
     aggr_production=e=
         sum(ownership,dom_consumption(ownership))+exports;

*Objective
objective..
     trick=e=1;

*Defining the model
MODEL labormarket /all/;

*Giving lower bounds to the variables
     production.lo("lowskilled")=0.00000000001 ;
     production.lo("highskilled")=0.00000000001 ;
     employment.lo("lowskilled")=0.00000000001;
     employment.lo("highskilled")=0.00000000001;
     aggr_profit.lo=0.00000000001;
     wage.lo("lowskilled")=0.00000000001;
     wage.lo("highskilled")=0.00000000001;
     goodsprices.lo("lowskilled")=0.00000000001;
     goodsprices.lo("highskilled")=0.00000000001;
     laborsupply.lo("lowskilled")=0.00000000001;
     laborsupply.lo("highskilled")=0.00000000001;
     unemployment.lo("highskilled")=0.00000000001;
     unemployment.lo("lowskilled")=0.00000000001;
     aggr_consumption.lo(ownership)=0.00000000001;
     importsprice.lo=0.00000000001;
     consumption_price.lo(ownership)=0.00000000001;
     productionprice.lo=0.00000000001;
     exports.lo=0.00000000001;
     imp_consumption.lo(ownership)=0.00000000001;
     dom_consumption.lo(ownership)=0.00000000001;
     taxallowance.lo(skill)=0.00000000001;

*Fixing import price to 1 and production in foreign country to a large number
    importsprice.fx=1.000;
    foreign_consumption.fx=1000000;

*Solving the model using NLP
SOLVE labormarket using NLP minimizing trick;

*Generating a table for checking the initial solution
*Variables in the table should be equal to zero
*For parameters the calibrated values are shown
set item0  /"Skilled producerwage", "Unskilled producerwage",
     "Skilled consumerwage", "Unskilled consumerwage", "Domestic price",

     "Consumption price", "Import price", "Skilled price",
     "Unskilled price", "Skilled employment", "Unskilled employment",
     "Skilled laborsupply", "Unskilled laborsupply", "Skilled unemployment",
     "Unskilled unemployment", "Skilled average tax",
     "Unskilled average tax", "Skilled benefit", "Unskilled benefit",
     "Production", "Skilled production", "Unskilled production",
     "Total consumption", "Private consumption", "Public consumption",
     "Exports", "Imports", "Imported public", "Imported private",
     "aggr_profit", "Skilled aggr_profit", "Unskilled aggr_profit",
     "Skilled replacement rate", "Unskilled replacement rate",
     "Skilled productivity", "Unskilled productivity", "Skilled d",
     "Unskilled d", "Skilled M", "Unskilled M", "Marginal tax",
     "Bargaining power", "epsilon lowskilled", "epsilon highskilled", 
     phi, b, "private q", "public q", x, ts, z, au, as/;

parameter report0(item0);
      report0("Skilled producerwage")
        =wage.l("highskilled")-wage0("highskilled");
      report0("Unskilled producerwage")
        =wage.l("lowskilled")-wage0("lowskilled");
      report0("Skilled consumerwage")
         =wage.l("highskilled")*(1-averagetax.l("highskilled"))
         -wage0("highskilled")*(1-averagetax0("highskilled"));
      report0("Unskilled consumerwage")
        =wage.l("lowskilled")*(1-averagetax.l("lowskilled"))
        -wage0("lowskilled")*(1-averagetax0("lowskilled"));
      report0("Domestic price")=productionprice.l-productionprice0;
      report0("Consumption price")
        =consumption_price.l("private")-consumptionprice0("private");
      report0("Import price")=importsprice.l-importsprice0;
      report0("Skilled price")
        =goodsprices.l("highskilled")-goodsprices0("highskilled");
      report0("Unskilled price")
        =goodsprices.l("lowskilled")-goodsprices0("lowskilled");
      report0("Skilled employment")
        =employment.l("highskilled")-employment0("highskilled");
      report0("Unskilled employment")
        =employment.l("lowskilled")-employment0("lowskilled");
      report0("Skilled laborsupply")
        =laborsupply.l("highskilled")-laborsupply0("highskilled");
      report0("Unskilled laborsupply")
        =laborsupply.l("lowskilled")-laborsupply0("lowskilled");
      report0("Skilled unemployment")
        =unemployment.l("highskilled")-unemployment0("highskilled");
      report0("Unskilled unemployment")
        =unemployment.l("lowskilled")-unemployment0("lowskilled");
      report0("Skilled average tax")
        =averagetax.l("highskilled")-averagetax0("highskilled");
      report0("Unskilled average tax")
        =averagetax.l("lowskilled")-averagetax0("lowskilled");
      report0("Skilled benefit")
        =benefits.l("highskilled")-benefits0("highskilled");
      report0("Unskilled benefit")
        =benefits.l("lowskilled")-benefits0("lowskilled");
      report0("Production")=aggr_production.l-aggr_production0;
      report0("Skilled production")
        =production.l("highskilled")-production0("highskilled");
      report0("Unskilled production")
        =production.l("lowskilled")-production0("lowskilled");
      report0("Total consumption")
        =sum(ownership,aggr_consumption.l(ownership))
        -sum(ownership,aggr_consumption0(ownership));
      report0("Private consumption")
        =aggr_consumption.l("private")-aggr_consumption0("private");
      report0("Public consumption")
        =aggr_consumption.l("public")-aggr_consumption0("public");
      report0("Exports")=exports.l-exports0;
      report0("Imports")

        =sum(ownership,imp_consumption.l(ownership))
        -sum(ownership,imp_consumption0(ownership));
      report0("Imported public")
        =imp_consumption.l("public")-imp_consumption0("public");
      report0("Imported private")
        =imp_consumption.l("private")-imp_consumption0("private");
      report0("aggr_profit")=aggr_profit.l-aggr_profit0;
      report0("Skilled aggr_profit")
        =profit0("highskilled")-profit.l("highskilled");
      report0("Unskilled aggr_profit")
        =profit0("lowskilled")-profit.l("lowskilled");
      report0("Skilled replacement rate")=replacementrate("highskilled");
      report0("Unskilled replacement rate")=replacementrate("lowskilled");
      report0("Skilled productivity")=productivity("highskilled");
      report0("Unskilled productivity")=productivity("lowskilled");
      report0("Skilled d")=d("highskilled");
      report0("Unskilled d")=d("lowskilled");
      report0("Skilled M")=m("highskilled");
      report0("Unskilled M")=m("lowskilled");
      report0("Marginal tax")=tm;
      report0("Bargaining power")=alpha;
      report0("epsilon lowskilled")=epsilon("lowskilled");
      report0("epsilon highskilled")=epsilon("highskilled");
      report0("phi")=phi;
      report0("private q")=q("private");
      report0("public q")=q("public");
      report0("b")=b;
      report0("x")=x;
      report0("ts")=ts;
      report0("z")=z;
      report0("au")=a("lowskilled");
      report0("as")=a("highskilled");

display  report0;

*$ontext
*Generating a table for reporting simulation results
set item3 /"Production", "Private consumption", "Public consumption", 
"Exports",
     "High-skilled producer wage",  "Low-skilled producer wage",
     "High-skilled consumer wage", "Low-skilled consumer wage",
     "Production price", "Consumption price", "High-skilled employment",
     "Low-skilled employment", "High-skilled labor supply",
     "Low-skilled labor supply", "High-skilled unemployment",
     "Low-skilled unemployment", "High-skilled replacement rate",
     "Low-skilled replacement rate", "High-skilled average tax",
     "Low-skilled average tax",  "Marginal tax", "Welfare"/;

*Setting up scenarios
set scenario  /1, 2, 3, 4, 5, 6/;

parameter report3(item3, scenario) Simulation results;
parameter decimals1(scenario) 2 decimals /"1" 2, "2" 2, "3" 2, "4" 2, "5" 2/;

set sim /s_alpha, s_riu, s_ris, s_fiu, s_fis/;

table scen(sim, scenario)
         1       2        3       4       5        6
s_alpha         0.181633 0.1822   0.1822  0.1822   0.1822
s_riu                    0.26987  0.2672  0.2672   0.2672
s_ris                    0.26987  0.26987 0.2672   0.2672
s_fiu                                     0.26577  0.26577
s_fis                                     0.18986  0.181
;

loop(scenario,

alpha$scen("s_alpha",scenario)=scen("s_alpha",scenario);

replacementrate("lowskilled")$scen("s_riu",scenario)=scen("s_riu",scenario);
replacementrate("highskilled")$scen("s_ris",scenario)=scen("s_ris",scenario);
a("lowskilled")$scen("s_fiu",scenario)=scen("s_fiu",scenario);
a("highskilled")$scen("s_fis",scenario)=scen("s_fis",scenario);

SOLVE labormarket using NLP minimizing trick;

     report3("Production", scenario)=aggr_production.l;
     report3("Private consumption", scenario)=aggr_consumption.l("private");
     report3("Public consumption", scenario)=aggr_consumption.l("public");
     report3("Exports", scenario)=exports.l;
     report3("High-skilled producer wage", scenario)
        =wage.l("highskilled")*(1+ts);
     report3("Low-skilled producer wage", 
scenario)=wage.l("lowskilled")*(1+ts);
     report3("High-skilled consumer wage", scenario)=wage.l("highskilled")*(1
        -(tm-tm*taxallowance.l("highskilled")*employment0("highskilled")
        /(wage.l("highskilled")*employment.l("highskilled"))));
     report3("Low-skilled consumer wage", scenario)=wage.l("lowskilled")*(1
        -(tm-tm*taxallowance.l("lowskilled")*employment0("lowskilled")
        /(wage.l("lowskilled")*employment.l("lowskilled"))));
     report3("Production price", scenario)=productionprice.l;
     report3("Consumption price", scenario)=consumption_price.l("private");
     report3("High-skilled employment", scenario)=employment.l("highskilled");
     report3("Low-skilled employment", scenario)=employment.l("lowskilled");
     report3("High-skilled labor supply", scenario)
        =laborsupply.l("highskilled");
     report3("Low-skilled labor supply", scenario)=laborsupply.l("lowskilled");
     report3("High-skilled unemployment", 
scenario)=laborsupply.l("highskilled")
        -employment.l("highskilled");
     report3("Low-skilled unemployment", scenario)=laborsupply.l("lowskilled")
        -employment.l("lowskilled");
     report3("High-skilled replacement rate", scenario)
        =replacementrate("highskilled");
     report3("Low-skilled replacement rate", scenario)
        =replacementrate("lowskilled");
     report3("High-skilled average tax", scenario)=tm
        -tm*taxallowance.l("highskilled")*employment0("highskilled")
        /(wage.l("highskilled")*employment.l("highskilled"));
     report3("Low-skilled average tax", scenario)=tm
        -tm*taxallowance.l("lowskilled")*employment0("lowskilled")
        /(wage.l("lowskilled")*employment.l("lowskilled"));
     report3("Marginal tax", scenario)=tm;
     report3("Welfare", scenario)=sum(skill,((d(skill)**(1/theta))
             *(((1-(tm-tm*taxallowance.l(skill)*employment0(skill)
             /(wage.l(skill)*employment.l(skill))))
             *wage.l(skill)/consumption_price.l("private"))**((theta-1)/theta))
             +((1-d(skill))**(1/theta))
             *(laborsupply.l(skill)/2)**(theta/(theta-1)))
             **(theta/(theta-1)))
             +aggr_consumption.l("public")+aggr_profit.l
             /consumption_price.l("private");
);

display  report3;

*Generating a table, which shows scenario results as percentage changes from
*the base solution
set column4  /1, 2, 3, 4, 5, 6/;

parameter report4(item3, column4) Simulation results;

report4(item3, "1")=100*(report3(item3,"2")
         -report3(item3,"1"))/report3(item3,"1");
report4(item3, "2")=100*(report3(item3,"3")
         -report3(item3,"1"))/report3(item3,"1");
report4(item3, "3")=100*(report3(item3,"4")

         -report3(item3,"1"))/report3(item3,"1");
report4(item3, "4")=100*(report3(item3,"5")
         -report3(item3,"1"))/report3(item3,"1");
report4(item3, "5")=100*(report3(item3,"6")
         -report3(item3,"1"))/report3(item3,"1");
display  report4 ;

* Generating a file with the table:
file vtex /results.tex/; put vtex;
$setglobal c_decimals decimals1
$setglobal title "Simulation results"
$setglobal title_just center
$setglobal format tex
$libinclude gams2tbl report4
putclose;
*$offtext

